```
(*calculation of equilibrium isotopic fractionation from NRIXS data; Dauphas 2011*)
Av = 6.02214179 × 10^23; (*Avogadro number*)
m4 = 53.9396105 × 10^-3 / Av; m6 = 55.9349375 × 10^-3 / Av;
m7 = 56.9353940 × 10^-3 / Av; (*mass of a Fe atoms in kg*)
k = 1.3806503 × 10^-23 ;(*Boltzmann constant m2kgs-2*)
h = 6.626068 × 10^-34 ;(*Planck constant*)
hb = h / (2 * Pi); (*hbar*)
Er = 1.956 * 0.001 * 1.60217646 × 10^-19; (*57Fe recoil energy J*)

datafpath =
 "/Users/dauphasu/Desktop/Jarosite_Goethite_GCA2011/sourcedata/data_Feb11copy/";
datafname = "Goethite";

Te = 300.8; (*temperature of the experiments in K for calculation of flm from g(E)*)

datafdos = FileNameJoin[{datafpath, StringJoin[datafname, ".dos"]}];
(*density of states*)
datafpsn = FileNameJoin[{datafpath, StringJoin[datafname, ".psn"]}];
(*excitation function S*)
Print[datafdos, "  FOUND:  ", FileExistsQ[datafdos]]
Print[datafpsn, "  FOUND:  ", FileExistsQ[datafpsn]]

mydatag = Import[datafdos, "Table"];
mydataS = Import[datafpsn, "Table", HeaderLines → 1];

(*plotting of S(E) and g(E)*)
ListPlot[Transpose[{mydataS[[All, 1]], mydataS[[All, 2]]}],
 AxesLabel → {"E (meV)", "S(E) (1/eV)"}, Joined → True]
ListPlot[Transpose[{mydatag[[All, 1]], mydatag[[All, 2]]}],
 AxesLabel → {"E (meV)", "g(E) (1/eV)"}, Joined → True]

(*Excitation function S(E)*)
(*the first column in E in meV, the second is S(E) in 1/eV,
and the third in the error of S(E) in 1/eV*)
EnS = mydataS[[All, 1]] * 0.001 * 1.60217646 × 10^-19; (*mEV to J*)
SE = mydataS[[All, 2]] / (1.60217646 × 10^-19); (*1/eV to 1/J*)
eS = mydataS[[All, 3]] / (1.60217646 × 10^-19); (*1/eV to 1/J*)
(*DOS g(E)*)
(*the first column in E in meV, the second is g(E) in 1/eV,
and the third in the error of g(E) in 1/eV*)
Eng = mydatag[[All, 1]] * 0.001 * 1.60217646 × 10^-19; (*mEV to J*)
gE = mydatag[[All, 2]] / (3 * 1.60217646 × 10^-19); (*1/eV to 1/J*)
eg = mydatag[[All, 3]] / (3 * 1.60217646 × 10^-19); (*1/eV to 1/J*)


g = Interpolation[Transpose[{Eng, gE}], InterpolationOrder → 1];
(*interpolated function*)
S = Interpolation[Transpose[{EnS, SE}], InterpolationOrder → 1];
(*interpolated function*)


flm = Exp[-Er * NIntegrate[(g[x] / x) * Coth[x / (2 * k * Te)],
    {x, Min[Eng] + 0.5 * 0.001 * 1.60217646 × 10^-19, Max[Eng]}]]

(*calculation of the nth order moment of the excitation function*)
mS[n_] := NIntegrate[S[x] × (x - Er)^n, {x, Min[EnS], Max[EnS]}];
(*numerical integration done by Mathematica on interpolated function*)
(*numerical integration using the Trapezoidal rule*)
mSt[n_] := 0.5 * Sum[(EnS[[i + 1]] - EnS[[i]]) *
    (SE[[i + 1]] * (EnS[[i + 1]] - Er)^n + SE[[i]] * (EnS[[i]] - Er)^n), {i, 1, Length[EnS] - 1}];
```

```
(*error propagation in the integration using the Rectangule rule*)
emSr[n_] :=
  √Sum[((EnS[[i + 1]] - EnS[[i]]) * ((eS[[i]]) * (EnS[[i]] - Er)^n))^2, {i, 1, Length[EnS] - 1}];
 (*error propagation in the integration using the Trapezoidal rule*)
emSt[n_] := 0.5 * (((EnS[[2]] - EnS[[1]]) * eS[[1]] * (EnS[[1]] - Er)^n)^2 +
       Sum[((EnS[[i + 1]] - EnS[[i - 1]]) * eS[[i]] * (EnS[[i]] - Er)^n)^2,
        {i, 2, Length[EnS] - 1}] + ((EnS[[Length[EnS]]] - EnS[[Length[EnS] - 1]]) *
          eS[[Length[EnS]]] * (EnS[[Length[EnS]]] - Er)^n)^2)^0.5;
(*moments of S*)
S2 = mS[2]; S2t = mSt[2]; eS2r = emSr[2]; eS2t = emSt[2];
S3 = mS[3]; S3t = mSt[3]; eS3r = emSr[3]; eS3t = emSt[3];
S4 = mS[4]; S4t = mSt[4]; eS4r = emSr[4]; eS4t = emSt[4];
S5 = mS[5]; S5t = mSt[5]; eS5r = emSr[5]; eS5t = emSt[5];
S6 = mS[6]; S6t = mSt[6]; eS6r = emSr[6]; eS6t = emSt[6];
S7 = mS[7]; S7t = mSt[7]; eS7r = emSr[7]; eS7t = emSt[7];

Gs2 = S3 / Er;
eGs2 = eS3t / Er;
Gs4 = (S5 - 10 * S2 * S3) / Er;
eGs4 = (eS5t^2 + 10^2 * S2^2 * eS3t^2 + 10^2 * S3^2 * eS2t^2)^0.5 / Er;
Gs6 = (S7 + 210 * S2^2 * S3 - 35 * S3 * S4 - 21 * S2 * S5) / Er;
eGs6 =
  (eS7t^2 + 210^2 * S2^4 * eS3t^2 + 210^2 * S3^2 * 4 * S2^2 * eS2t^2 + 35^2 * S3^2 * eS4t^2 +
      35^2 * S4^2 * eS3t^2 + 21^2 * S2^2 * eS5t^2 + 21^2 * S5^2 * eS2t^2)^0.5 / Er;


(*force constant in N/m from S(E)*)
Fs = m7 / (Er * hb^2) * S3;
eFs = m7 / (Er * hb^2) * eS3t;

(*calculation of the nth order moments of the PDOS*)
mG[n_] := NIntegrate[g[x] × x^n, {x, Min[Eng], Max[Eng]}]
(*integration using the interpolated function*)
mGt[n_] :=
 0.5 * Sum[(Eng[[i + 1]] - Eng[[i]]) * (gE[[i + 1]] * (Eng[[i + 1]])^n + gE[[i]] * (Eng[[i]])^n),
   {i, 1, Length[Eng] - 1}] (*numerical integration using the Trapezoidal method*)
emGr[n_] := √Sum[((Eng[[i + 1]] - Eng[[i]]) * ((eg[[i]]) * (Eng[[i]])^n))^2,
   {i, 1, Length[Eng] - 1}]
   (*error propagation in the integration using the Rectangular method*)
emGt[n_] := 0.5 * (((Eng[[2]] - Eng[[1]]) * eg[[1]] * (Eng[[1]])^n)^2 +
      Sum[((Eng[[i + 1]] - Eng[[i - 1]]) * eg[[i]] * (Eng[[i]])^n)^2, {i, 2, Length[Eng] - 1}] +
      ((Eng[[Length[Eng]]] - Eng[[Length[Eng] - 1]]) *
         eg[[Length[Eng]]] * (Eng[[Length[Eng]]])^n)^2)^0.5
(*error propagation in the integration using the Trapezoidal method*)

(*2nd, 4th, and 6th orders*)
G2 = mG[2]; G2t = mGt[2]; eG2r = emGr[2]; eG2t = emGt[2];
G4 = mG[4]; G4t = mGt[4]; eG4r = emGr[4]; eG4t = emGt[4];
G6 = mG[6]; G6t = mGt[6]; eG6r = emGr[6]; eG6t = emGt[6];

(*calculation of the force constant and the coefficients of the polynomial*)
Fg = m7 / hb^2 G2; (*force constant from PDOS in N/m*)
eFg = m7 / hb^2 * eG2t; (*uncertainty on force constant*)
```

```
Print["Force constant from PDOS= ", Fg,
 "±", eFg, "; from S(E)= ", Fs, "±", eFs, " (N/m)"]
(*the following print out compares the 2n moment of g with that calculated from S*)
Print["G2/Gs2s: ", G2 / Gs2, "; G4/Gs4: ", G4 / Gs4, "; G6/Gs6: ", G6 / Gs6]
D1g = 1000 × (m7 (1 / m4 - 1 / m6))/(8 × k^2) × G2; eD1g = 1000 × (m7 (1 / m4 - 1 / m6))/(8 × k^2) × eG2t;
D2g = -1000 × (m7 (1 / m4 - 1 / m6))/(480 × k^4) × G4; eD2g = 1000 × (m7 (1 / m4 - 1 / m6))/(480 × k^4) × eG4t;
D3g = 1000 × (m7 (1 / m4 - 1 / m6))/(20 160 × k^6) × G6; eD3g = 1000 × (m7 (1 / m4 - 1 / m6))/(20 160 × k^6) × eG6t;
D1S = 1000 × (m7 (1 / m4 - 1 / m6))/(8 × k^2) × Gs2; eD1S = 1000 × (m7 (1 / m4 - 1 / m6))/(8 × k^2) × eGs2;
D2S = -1000 × (m7 (1 / m4 - 1 / m6))/(480 × k^4) × Gs4; eD2S = 1000 × (m7 (1 / m4 - 1 / m6))/(480 × k^4) × eGs4;
D3S = 1000 × (m7 (1 / m4 - 1 / m6))/(20 160 × k^6) × Gs6; eD3S = 1000 × (m7 (1 / m4 - 1 / m6))/(20 160 × k^6) × eGs6;


Print["from g(E): D1=", D1g ± eD1g, "; D2=", D2g ± eD2g, "; D3=", D3g ± eD3g]
Print["from S(E): D1=", D1S ± eD1S, "; D2=", D2S ± eD2S, "; D3=", D3S ± eD3S]

tmin = 0; tmax = 500; tstep = 10; (*temperature in celsius*)
(*calculates the beta values using the full
 integral for temperatures tmin to tmax in increments tstep*)
betag = Table[{i, 1000 × m7 (1 / m4 - 1 / m6) * 1.5 *
     (NIntegrate[g[x] × ((x / (k * (i + 273.15)))/(e^(x/(k*(i+273.15))) - 1) + 0.5 * x / (k * (i + 273.15)) - 1),
        {x, Min[Eng], Max[Eng]}])}, {i, tmin, tmax, tstep}];
betaS = Table[{i, D1S / (i + 273.15) ^2 + D2S / (i + 273.15) ^4 + D3S / (i + 273.15) ^6},
   {i, tmin, tmax, tstep}];


(*calculation of the coefficients in the polynomial 1000 ln beta=
 B1<F>/T^2+B2<F>^2/T^4 with <F> in N/m and T  in K*)
B1 = hb^2 * 1000 * (1 / m4 - 1 / m6) / (8 * k^2);
betares = betaS;
betares[[All, 2]] = betares[[All, 2]] - D1S / (betares[[All, 1]] + 273.15) ^2;
B2 = -Numerator[Fit[betares, {1 / (x + 273.15) ^4}, x]] / Fs^2;
Print["B1=", B1, "; B2=", B2]
Show[Plot[{B1 * Fs / (x + 273.15)^2 - B2 * Fs^2 / (x + 273.15)^4},
  {x, 0, 100}, AxesLabel → {"T (K)", "1000 ln β"}], ListPlot[betaS]]
Show[Plot[{{D1g / (x + 273.15)^2 + D2g / (x + 273.15)^4 + D3g / (x + 273.15)^6},
   {D1S / (x + 273.15)^2 + D2S / (x + 273.15)^4 + D3S / (x + 273.15)^6}},
  {x, 0, 100}, AxesLabel → {"T (K)", "1000 ln β"}], ListPlot[betaS]]
/Users/dauphasu/Desktop/Jarosite_Goethite_GCA2011/sourcedata/data_Feb11copy/Goethite.dos
   FOUND:  True
/Users/dauphasu/Desktop/Jarosite_Goethite_GCA2011/sourcedata/data_Feb11copy/Goethite.psn
   FOUND:  True
```





0.773692

Force constant from PDOS= 316.334±10.8674; from S(E)= 307.415±9.40995 (N/m)

G2/Gs2s: 1.02901; G4/Gs4: 1.10801; G6/Gs6: 1.41387

from g(E): D_1=918 787. ± 31 564.1; D_2=
$-8.1491\times10^{9}\pm8.29987\times10^{8}$; D_3=$1.9636\times10^{14}\pm3.59325\times10^{13}$

from S(E): D_1=892 881. ± 27 331.; D_2=
$-7.35471\times10^{9}\pm6.92952\times10^{8}$; D_3=$1.38881\times10^{14}\pm2.97415\times10^{13}$

B_1=2904.48; B_2=61 951.3




1000 ln β
11
10
9
8
7
20
40
60
80
100
T (K)